# $Id: PKGBUILD 59626 2011-11-28 21:56:45Z giovanni $
# Maintainer: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: suasageandeggs <s_stoakley@hotmail.co.uk>
# Contributor: Michael Fellinger <manveru@www.weez-int.com>

pkgname=libtommath
pkgver=0.42.0
pkgrel=2
pkgdesc="Highly optimized and portable routines for integer based number theoretic applications"
url="http://www.libtom.org/"
license=('custom')
arch=('i686' 'x86_64')
source=("http://libtom.org/files/ltm-${pkgver}.tar.bz2"
        'libtommath-0.42.0-makefile.patch')
md5sums=('7380da904b020301be7045cb3a89039b'
         'd50c26fdc5aa87e6acfd4adc20557ff4')

build() {
  cd $srcdir/$pkgname-$pkgver

  patch -Np1 -i ${srcdir}/libtommath-0.42.0-makefile.patch

  make -f makefile.shared IGNORE_SPEED=1
}

package() {
  cd $srcdir/$pkgname-$pkgver

  make DESTDIR=$pkgdir INSTALL_GROUP=root install
  install -Dm644 LICENSE "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
  chmod 644 ${pkgdir}/usr/lib/libtommath.a
}
